6ème Journée thématique : Modélisation - Simulation

MODELICA® : a modeling language for dynamic simulation of multi-domain complex physical systems

Dr Ir Eric Dumont

Faculty of Engineering – Thermodynamics Department

Mons, January 28th 2016

UMONS RESEARCH INSTITUTE FOR ENERGY

Introduction (1/5)

- **Modelica**® is a free language developed by the non-profit Modelica Association (https://www.modelica.org/)
- Development started in 1996 in an effort to develop an object-oriented language for modeling of technical systems in order to reuse and exchange dynamic system models in a standardized format
- Based on PhD Thesis of H. Elmqvist, 1978, University of Lund, Sweden
- Release of Modelica 1.0 in 1997
- Release of Modelica 3.3 in 2012
- Several European Projects helped developing the **Modelica**® language

Introduction (2/5)

UMONS RESEARCH INSTITUTE FOR ENERGY

- **Modelica**® is a non-proprietary, object-oriented, equation-based language to conveniently model complex physical systems containing, e.g., mechanical, electrical, electronic, hydraulic, thermal, control, electric power or process-oriented subcomponents
- It can use many components described in libraries :

- Modelica Standard Library (Modelica Association, 1280 model components and 910 fonctions)

- 81 free libraries (January 2016)

- 29 commercial libraries (January 2016)

Introduction (3/5)

- **Equation-based**

- models are described by a set of equations (mathematical description) and not algorithm blocks (like standard programming languages C++, FORTRAN, MATLAB). That means there is no causality in the equations

- equations are algebraic equations, ordinary differential equations and discrete equations, NOT partial differential equations

Introduction (4/5)

energy
UMONS RESEARCH INSTITUTE FOR ENERGY

- **Object-oriented**

- actually “component-oriented” so that each “physical” component can be reused by other people, just “connecting” components together to develop a new model
- well-suited for system-level simulation
- used by the automotive industry (Ford, General Motors, Toyota, BMW, Audi, etc.)
- used by many universities in the world (mainly in the USA and Europe)

Introduction (5/5)

- **Modeling language**

- equation set is not compiled but translated into objects and then run in a simulation environment (which is not part of the **Modelica**® language)

- the simulation environment has several algorithms to reorganize the equations and to solve them

- **Modelica**® simulation environments :

- free : e.g. OpenModelica (Linköping University)

- commercial : e.g. DYMOLA (Dassault Systèmes)

MODELICA® language (1/8)

UMONS RESEARCH INSTITUTE FOR ENERGY

- All examples are from “Modelica by Example” (http://book.xogeny.com/)
- Simple first-order model

$$\dot{x} = (1 - x)$$

```
model FirstOrderDocumented "A simple first order differential equation"
  Real x "State variable";
equation
  der(x) = 1-x "Drives value of x toward 1.0";
end FirstOrderDocumented;
```



MODELICA® language (2/8)

energy
UMONS RESEARCH INSTITUTE FOR ENERGY

- Simple first-order model (continued)

$$\dot{x} = (1 - x)$$

```
model FirstOrderInitial "First order equation with initial value"
  Real x "State variable";
initial equation
  x = 2 "Used before simulation to compute initial values";
equation
  der(x) = 1-x "Drives value of x toward 1.0";
end FirstOrderInitial;
```




```
model FirstOrderSteady
  "First order equation with steady state initial condition"
  Real x "State variable";
initial equation
  der(x) = 0 "Initialize the system in steady state";
equation
  der(x) = 1-x "Drives value of x toward 1.0";
end FirstOrderSteady;
```



MODELICA® language (3/8)

energy
UMONS RESEARCH INSTITUTE FOR ENERGY

- Simple electrical circuit

$$V = i_R R$$

$$C\frac{\mathrm{d}V}{\mathrm{d}t} = i_C$$

$$L\frac{\mathrm{d}i_L}{\mathrm{d}t} = (V_b - V)$$

$$i_L = i_R + i_C$$




```
model RLC1 "A resistor-inductor-capacitor circuit model"
  type Voltage=Real(unit="V");
  type Current=Real(unit="A");
  type Resistance=Real(unit="Ohm");
  type Capacitance=Real(unit="F");
  type Inductance=Real(unit="H");
  parameter Voltage Vb=24 "Battery voltage";
  parameter Inductance L = 1;
  parameter Resistance R = 100;
  parameter Capacitance C = 1e-3;
  Voltage V;
  Current i_L;
  Current i_R;
  Current i_C;
equation
  V = i_R*R;
  C*der(V) = i_C;
  L*der(i_L) = (Vb-V);
  i_L=i_R+i_C;
end RLC1;
```



MODELICA® language (4/8)

- Switched electrical circuit

```
model SwitchedRLC "An RLC circuit with a switch"
  type Voltage=Real(unit="V");
  type Current=Real(unit="A");
  type Resistance=Real(unit="Ohm");
  type Capacitance=Real(unit="F");
  type Inductance=Real(unit="H");
  parameter Voltage Vb=24 "Battery voltage";
  parameter Inductance L = 1;
  parameter Resistance R = 100;
  parameter Capacitance C = 1e-3;
  Voltage Vs;
  Voltage V;
  Current i_L;
  Current i_R;
  Current i_C;
equation
  Vs = if time>0.5 then Vb else 0;
  i_R = V/R;
  i_C = C*der(V);
  i_L=i_R+i_C;
  L*der(i_L) = (Vs-V);
end SwitchedRLC;
```





MODELICA® language (5/8)

energy
UMONS RESEARCH INSTITUTE FOR ENERGY

- One-dimensional heat transfer



$$\rho A L_i C \frac{\mathrm{d}T_1}{\mathrm{d}t} = -kA\frac{T_1 - T_2}{L_i} - hA(T_1 - T_{amb})$$

$$\rho A L_i C \frac{\mathrm{d}T_n}{\mathrm{d}t} = -kA\frac{T_n - T_{n-1}}{L_i} - hA(T_n - T_{amb})$$

$$\rho A L_i C \frac{\mathrm{d}T_i}{\mathrm{d}t} = -kA\frac{T_i - T_{i-1}}{L_i} - kA\frac{T_i - T_{i+1}}{L_i}$$




```
model Rod_ForLoop "Modeling heat conduction in a rod using a for loop"
  type Temperature=Real(unit="K", min=0);
  type ConvectionCoefficient=Real(unit="W/K", min=0);
  type ConductionCoefficient=Real(unit="W.m-1.K-1", min=0);
  type Mass=Real(unit="kg", min=0);
  type SpecificHeat=Real(unit="J/(K.kg)", min=0);
  type Density=Real(unit="kg/m3", min=0);
  type Area=Real(unit="m2");
  type Volume=Real(unit="m3");
  type Length=Real(unit="m", min=0);
  type Radius=Real(unit="m", min=0);

  constant Real pi = 3.14159;

  parameter Integer n=10;
  parameter Length L=1.0;
  parameter Radius R=0.1;
  parameter Density rho=2.0;
  parameter ConvectionCoefficient h=2.0;
  parameter ConductionCoefficient k=10;
  parameter SpecificHeat C=10.0;
  parameter Temperature Tamb=300 "Ambient temperature";

  parameter Area A = pi*R^2;
  parameter Volume V = A*L/n;

  Temperature T[n];
initial equation
  T = linspace(200,300,n);
equation
  rho*V*C*der(T[1]) = -h*(T[1]-Tamb)-k*A*(T[1]-T[2])/(L/n);
  for i in 2:(n-1) loop
    rho*V*C*der(T[i]) = -k*A*(T[i]-T[i-1])/(L/n)-k*A*(T[i]-T[i+1])/(L/n);
  end for;
  rho*V*C*der(T[end]) = -h*(T[end]-Tamb)-k*A*(T[end]-T[end-1])/(L/n);
end Rod_ForLoop;
```

MODELICA® language (6/8)

energy
UMONS RESEARCH INSTITUTE FOR ENERGY

- Components : connectors

- “Across” variables (potential)
- “Through” variables (conserved quantity)

```
connector Thermal
   Modelica.SIunits.Temperature T;
   flow Modelica.SIunits.HeatFlowRate q;
end Thermal;
```

$T_A = T_B$

$q_A + q_B = 0$

```
connector Electrical
   Modelica.SIunits.Voltage v;
   flow Modelica.SIunits.Current i;
end Electrical;
```

$V_A = V_B$

$I_A + I_B = 0$

Domain	Through Variable	Across Variable
Electrical	Current [A]	Voltage [V]
Thermal	Heat [W]	Temperature [K]
Translational	Force [N]	Position [m]
Rotational	Torque [N.m]	Angle [rad]

MODELICA® language (7/8)

energy
UMONS RESEARCH INSTITUTE FOR ENERGY

- Components : use of connectors

```
within ModelicaByExample.Components.HeatTransfer;
model ThermalCapacitance "A model of thermal capacitance"
  parameter Modelica.SIunits.HeatCapacity C "Thermal capacitance";
  parameter Modelica.SIunits.Temperature T0 "Initial temperature";
  Modelica.Thermal.HeatTransfer.Interfaces.HeatPort_a node
    annotation (Placement(transformation(extent={{-10,-10},{10,10}})));
initial equation
  node.T = T0;
equation
  C*der(node.T) = node.Q_flow;
end ThermalCapacitance;
```





```
within ModelicaByExample.Components.HeatTransfer;
model ConvectionToAmbient "An overly specialized model of convection"
  parameter Modelica.SIunits.CoefficientOfHeatTransfer h;
  parameter Modelica.SIunits.Area A;
  parameter Modelica.SIunits.Temperature T_amb "Ambient temperature";
  Modelica.Thermal.HeatTransfer.Interfaces.HeatPort_a port_a
    annotation (Placement(transformation(extent={{-110,-10},{-90,10}})));
equation
  port_a.Q_flow = h*A*(port_a.T-T_amb) "Heat transfer equation";
end ConvectionToAmbient;
```

```
within ModelicaByExample.Components.HeatTransfer;
model AmbientCondition
  "Model of an \"infinite reservoir\" at some ambient temperature"
  parameter Modelica.SIunits.Temperature T_amb "Ambient temperature";
  Modelica.Thermal.HeatTransfer.Interfaces.HeatPort_a node annotation (
      Placement(transformation(extent={{-10,-10},{10,10}}), iconTransformation(
          extent={{-10,-10},{10,10}})));
equation
  node.T = T_amb;
end AmbientCondition;
```



MODELICA® language (8/8)

- Components : graphical system model





Stratified water tank (1/6)

energy
UMONS RESEARCH INSTITUTE FOR ENERGY

- Model of heat transfer in a stratified water tank heated by a heat pump





Stratified water tank (2/6)

- Heat transfer and energy equations for a standard layer

(134 layers have been used)

$$\underbrace{\rho \cdot c_p \cdot V_i \cdot \frac{dT_i}{dt}}_{\text{Variation de T° de la strate}} = \underbrace{-U \cdot A_i \cdot (T_i - T_{ext})}_{\text{Pertes vers l'ext.}} + \underbrace{\frac{k}{h} \cdot A_s \cdot (T_{i-1} - T_i)}_{\text{Flux échangé avec la strate du haut}} - \underbrace{\frac{k}{h} \cdot A_s \cdot (T_i - T_{i+1})}_{\text{Flux échangé avec la strate du bas}} + \underbrace{\phi_{ci}}_{\text{Flux reçu du condenseur}} + \underbrace{q_m \cdot c_p \cdot (T_{i+1} - T_i)}_{\text{Eau chaude sanitaire consommée}} \qquad (8)$$



Stratified water tank (3/6)

- Temperature profile in water tank

(begin and end of one heating cycle) for k = 40 W/(m.K)



Stratified water tank (4/6)

- Temperature profile in water tank for three layers (bottom, center and top)

k = 40 W/(m.K)





k = 20000 W/(m.K)

Stratified water tank (5/6)

- Temperature profile in water tank for the layer with the temperature sensor compared with measurements

(one heating cycle)



Stratified water tank (6/6)

energy
UMONS RESEARCH INSTITUTE FOR ENERGY

- Model of heat transfer in a stratified water tank heated by solar panels





Conclusions

energy
UMONS RESEARCH INSTITUTE FOR ENERGY

- **Modelica**® is a powerful language for dynamic simulation of complex multi-domain physics systems

- Equation-based
- Component-oriented
- Graphical building of systems
- Many available components and libraries

That's All Folks !